# Maintainer: Nathan Osman <nathan@quickmediasolutions.com>

_realname=qmdnsengine
pkgbase=mingw-w64-${_realname}
pkgname="${MINGW_PACKAGE_PREFIX}-${_realname}"
pkgver=0.2.0
pkgrel=1
pkgdesc="Multicast DNS library for Qt applications"
arch=('any')
license=('MIT')
url='https://github.com/nitroshare/qmdnsengine'
depends=("${MINGW_PACKAGE_PREFIX}-qt5>=5.4")
makedepends=("${MINGW_PACKAGE_PREFIX}-cmake>=3.2"
             "${MINGW_PACKAGE_PREFIX}-gcc")
source=("https://github.com/nitroshare/qmdnsengine/archive/${pkgver}.tar.gz")
sha256sums=('b9a4e86de2afd31224ca2007713a2d9bcc0334358f831a2b83a7161b7d91c614')

build() {
  mkdir ${srcdir}/build-${MINGW_CHOST}
  cd ${srcdir}/build-${MINGW_CHOST}
  MSYS2_ARG_CONV_EXCL="-DCMAKE_INSTALL_PREFIX=" \
  ${MINGW_PREFIX}/bin/cmake \
    -G"MSYS Makefiles" \
    -DCMAKE_INSTALL_PREFIX=${MINGW_PREFIX} \
    -DCMAKE_BUILD_TYPE=Release \
    ${srcdir}/${_realname}-${pkgver}
  make
}

package() {
  cd ${srcdir}/build-${MINGW_CHOST}
  make DESTDIR="${pkgdir}" install
}
